

## 5. 偏微分方程式の解析

### 5.1 偏微分方程式の厳密解

```
> restart;  
with(PDEtools):
```

熱伝導方程式の厳密解を求めます。  
式を作成します。

```
> PDE:=diff(u(x,t),t)=k*diff(u(x,t),x$2);
```

pdsolve 関数を利用して偏微分方程式を解きます。

```
> ans:=pdsolve(PDE,u(x,t),build);
```

予約語 \_c の定義を解除します。

```
> unprotect(_c);
```

任意定数 \_c[1],\_C1,\_C2,\_C3 に値を割り当てます。

```
> _c[1]:=0.8:_C1:=1/2:_C2:=0.1:_C3:=1.2:
```

定数 k にも値を割り当てます。

```
> k:=0.3:
```

```
> ans;
```

時間 t が 0 ~ 5 までをアニメーションで表示します。

```
> plots[animate](rhs(ans),x=0..1,t=0..5);
```

3次元グラフィックスで表示します。

```
> plot3d(rhs(ans),x=0..1,t=0..5,axes=box,orientation=[-60,60]);
```

### 5.2 偏微分方程式の数値解

波動方程式の数値解を求めます。

```
> restart:
```

式を作成します。

```
> PDE := diff(u(x,t),t)=-diff(u(x,t),x);
```

初期条件の定義

```
> IBC := {u(x,0) = exp(x)*sin(2*Pi*x), u(0,t) = -sin(2*Pi*t)/exp  
(t)};
```

pdsolve 関数を利用し、オプション numeric を用いることで数値解として求めます。

```
> ANS:=pdsolve(PDE,IBC,numeric,time=t,range=0..1,spacestep=1/256);
```

t=0,t=0.5,t=1 の時のグラフィックスを一緒に表示します。

```
> p1:=ANS:-plot(t=0,legend="t=0");
```

```
p2:=ANS:-plot(t=1/2,color=blue,legend="t=0.5");
```

```
p3:=ANS:-plot(t=1,color=green,legend="t=1");
```

```
plots[display]([p1,p2,p3]);
```

t=1 までをアニメーションで表示します。

```
> ANS:-animate(t=5,frames=100);
```